
Warranty and Assistance
The CS547 CONDUCTIVITY AND TEMPERATURE PROBE AND A547
INTERFACE are warranted by CAMPBELL SCIENTIFIC, INC. to be free
from defects in materials and workmanship under normal use and service for
twelve (12) months from date of shipment unless specified otherwise.  Batteries
have no warranty.  CAMPBELL SCIENTIFIC, INC.'s obligation under this
warranty is limited to repairing or replacing (at CAMPBELL SCIENTIFIC,
INC.'s option) defective products.  The customer shall assume all costs of
removing, reinstalling, and shipping defective products to CAMPBELL
SCIENTIFIC, INC.  CAMPBELL SCIENTIFIC, INC. will return such
products by surface carrier prepaid.  This warranty shall not apply to any
CAMPBELL SCIENTIFIC, INC. products which have been subjected to
modification, misuse, neglect, accidents of nature, or shipping damage.  This
warranty is in lieu of all other warranties, expressed or implied, including
warranties of merchantability or fitness for a particular purpose.  CAMPBELL
SCIENTIFIC, INC. is not liable for special, indirect, incidental, or
consequential damages.

Products may not be returned without prior authorization.  The following
contact information is for US and International customers residing in countries
served by Campbell Scientific, Inc. directly.  Affiliate companies handle repairs
for customers within their territories.  Please visit www.campbellsci.com to
determine which Campbell Scientific company serves your country.  To obtain
a Returned Materials Authorization (RMA), contact CAMPBELL
SCIENTIFIC, INC., phone (435) 753-2342.  After an applications engineer
determines the nature of the problem, an RMA number will be issued.  Please
write this number clearly on the outside of the shipping container.
CAMPBELL SCIENTIFIC's shipping address is:

CAMPBELL SCIENTIFIC, INC.
RMA#_____
815 West 1800 North
Logan, Utah 84321-1784

CAMPBELL SCIENTIFIC, INC. does not accept collect calls.

CS547 Conductivity and Temperature
Probe and A547 Interface

1.  Overview
The CS547 conductivity and temperature probe, and A547 interface are designed
for measuring the electrical conductivity, dissolved solids, and temperature of
fresh water with Campbell Scientific dataloggers.  They require the use of AC
excitation, so they can be used with the CR10(X), 21X, and CR7 dataloggers but
not with the BDR301 or BDR320.  Use with multiplexers is possible.

Electrical conductivity (EC) of a solution is a simple physical property, but
measurements can be difficult to interpret.  This manual instructs the user how to
make EC measurements with the CS547.  Accuracy specifications apply to
measurements of EC in water containing KCl, Na2SO4, NaHCO3, and/or NaCl,
which are typical calibration compounds, and to EC not yet compensated for
temperature effects.

Statements made on methods of temperature compensation or estimating dissolved
solids are included to introduce common ways of refining and interpreting data, but
are not definitive.  Authoritative sources to consult include the USGS Water-Supply
Paper 1473, The pH and Conductivity Handbook published by OMEGA
Engineering, physical chemistry texts, and other sources.

1.1  EC Sensor
The EC sensor consists of three stainless steel rings mounted in an epoxy tube
as shown in Figure 4-1.  Resistance of water in the tube is measured by
excitation of the center electrode with positive and negative voltage.

This electrode configuration eliminates the ground looping problems associated
with sensors in electrical contact with earth ground.

Temperature is measured with a thermistor in a three wire half bridge
configuration.

1.2  A547 Interface
The interface contains the completion resistors and blocking capacitors.  The
interface should be kept in a non-condensing environment that is maintained
within the temperature range of the unit.

FIGURE 1-1.  A547 Interface and CS547 Conductivity and Temperature Probe

2.  Specifications

2.1  CS547 Probe
Construction The probe housing is stainless steel

Size Probe Length:  3.7 inches (94 mm)
Diameter:  0.95 inches (24.13 mm)

Maximum Cable
Length

1000 ft.  The sensor must be ordered with desired length
as cable cannot be added to existing probes.

Depth Rating Maximum 1000 feet

pH Range Solution pH of less than 3.0 or greater than 9.0 may damage
the stainless steel housing.

Electrodes Passivated 316 SS with DC isolation capacitors.

Cell Constant Individually calibrated.  The cell constant (Kc) is found on
a label near the termination of the cable.

Temp. Range of Use 0° to 50°C.

EC Range Approx. 0.005 to 7.0 mS cm-1.

Accuracy in KCl and Na2SO4, NaHCO3, and NaCl standards at
25°C:

±5% of reading 0.44 to 7.0 mS cm-1.
±10% of reading 0.005 to 0.44 mS cm-1.

2.2  A547 Interface
Size Dimensions:  2.5” x .875” x 1.750

Weight:  6 oz.

Temperature Rating -15°C to +50°C

2.3  Temperature Sensor
Thermistor Betatherm 100K6A1.

Range 0°C to 50°C.

Accuracy Error ±0.4°C (See Section 8.2).

3.  Installation
Rapid heating and cooling of the probe, such as leaving it
in the sun and then submersing it in a cold stream, may
cause irreparable damage.

3.1  Site Selection
The EC sensor measures the EC of water inside the stainless steel tube, so
detection of rapid changes in EC requires that the probe be flushed
continuously.  This is easy to accommodate in a flowing stream by simply
orienting the sensor parallel to the direction of flow.  In stilling wells and
ground wells, however, diffusion rate of ions limits the response time.

3.2  Mounting
The stainless steel housing and sensor cable are made of water impervious,
durable materials.  Care should be taken, however, to mount the probe where
contact with abrasives and moving objects will be avoided.  Strain on cables
can be minimized by using a split mesh strain relief sleeve on the cable, which
is recommended for cables over 100 ft.  The strain relief sleeve is available
from Campbell Scientific as part number 7421.

The A547 is usually mounted in the datalogger enclosure.

4.  Wiring
The excitation channel used for EC must be separate
from the one used for temperature or measurement
errors will result.

CAUTION

WARNING

FIGURE 4-1.  CS547 Wiring Diagram for Example Below

5.  Programming

5.1  Programming Overview
Typical datalogger programs to measure the CS547 consist of four parts:

1. Measurement of EC and temperature

2. Correction of ionization errors in EC measurements

3. Correction of temperature errors in EC measurements

4. Output processing

All example programs may require modification by the user to fit the specific
application's wiring and programming needs.  All example programs in this
manual are for the CR10(X) and assume that datalogger wire connections are as
follows:  the LO COND lead is connected to 1L, the HI COND to 1H, the EX
COND to EX1, the EX TEMP to EX2, and the SE TEMP to SE3.

5.2  Measurement Programming.
EC:  Results from Instructions 5 or 6 (chosen automatically as part of the
autoranging feature of the following program segment) are processed with
Instruction P59 to produce the resistance across the electrodes:

Input Location Labels

Definitions for the following program:
Rs Solution resistance
Rp Resistance of leads/cable and blocking caps
Ct Solution EC with no temp. correction
C25mScm_1 EC corrected for temperature

AG
SE3
EX2
EX1
1H

1L

Clear (Shield)

Red (Temp)
Orange (Cond)
Black (Ex Cond)
Green (Ex Temp)

*Table 1 Program
01: 5 Execution Interval (seconds)

;Make a preliminary measurement of resistance for autoranging.

1:  Full Bridge (P6)
1: 1 Reps
2: 15 ±2500 mV Fast Range
3: 1 DIFF Channel
4: 1 Excite all reps w/Exchan 1
5: 2500 mV Excitation
6: 1 Loc [ Rs        ]
7: -.001 Mult
8: 1 Offset

2:  BR Transform Rf [X/(1-X)] (P59)
1: 1 Reps
2: 1 Loc [ Rs        ]
3: 1 Multiplier (Rf)

;
;Test the initial measurement to make a more accurate measurement.
;

3:  CASE (P93)
1: 1 Case Loc [ Rs        ]

4:  If Case Location < F (P83)
1: 1.8 F
2: 30 Then Do

5:  AC Half Bridge (P5)
1: 1 Reps
2: 15 ±2500 mV Fast Range
3: 2 SE Channel
4: 1 Excite all reps w/Exchan 1
5: 2500 mV Excitation
6: 1 Loc [ Rs        ]
7: 1.0 Mult
8: 0.0 Offset

6:  BR Transform Rf[X/(1-X)] (P59)
1: 1 Reps
2: 1 Loc [ Rs        ]
3: 1 Multiplier (Rf)

7:  End (P95)

8:  If Case Location < F (P83)
1: 9.25 F
2: 30 Then Do

(cont.)

9:  Full Bridge (P6)
1: 1 Reps
2: 15 ±2500 mV Fast Range
3: 1 DIFF Channel
4: 1 Excite all reps w/Exchan 1
5: 2500 mV Excitation
6: 1 Loc [ Rs        ]
7: -.001 Mult
8: 1 Offset

10:  BR Transform Rf[X/(1-X)] (P59)
1: 1 Reps
2: 1 Loc [ Rs        ]
3: 1 Multiplier (Rf)

11:  End (P95)

12:  If Case Location < F (P83)
1: 280 F
2: 30 Then Do

13:  Full Bridge (P6)
1: 1 Reps
2: 14 ±250 mV Fast Range
3: 1 DIFF Channel
4: 1 Excite all reps w/Exchan 1
5: 2500 mV Excitation
6: 1 Loc [ Rs        ]
7: -.001 Mult
8: 1 Offset

14:  BR Transform Rf[X/(1-X)] (P59)
1: 1 Reps
2: 1 Loc [ Rs        ]
3: 1 Multiplier (Rf)

15:  End (P95)

16:  End (P95)

;
;Subtract resistance errors (Rp) caused by the blocking capacitors
;(0.005Kohm) and the cable length (0.000032kohm/ft).  Enter cable lead
;length in nnn below.
;

17:  Z=F (P30)
1: nnn F Enter cable length in feet.
2: 00 Exponent of 10
3: 5 Z Loc [ Rp        ]

18:  Z=X*F (P37)
1: 5 X Loc [ Rp        ]
2: .00032 F
3: 5 Z Loc [ Rp        ]

(cont.)

19:  Z=X*F (P37)
1: 5 X Loc [ Rp        ]
2: -.1 F
3: 5 Z Loc [ Rp        ]

20:  Z=X+F (P34)
1: 5 X Loc [ Rp        ]
2: -.005 F
3: 5 Z Loc [ Rp        ]

21:  Z=X+Y (P33)
1: 1 X Loc [ Rs        ]
2: 5 Y Loc [ Rp        ]
3: 1 Z Loc [ Rs        ]

;
;EC is then calculated by multiplying the reciprocal of resistance,
;which is conductance, by the cell constant.
;

NOTE:  The cell constant (Kc) is printed on the label of each sensor or it can be calculated
(see Section 6.4).  It is entered in place of nnn below.

22:  Z=1/X (P42)
1: 1 X Loc [ Rs        ]
2: 2 Z Loc [ l_over_Rs ]

23:  Z=X*F (P37)
1: 2 X Loc [ l_over_Rs ]
2: nnn F Enter cell constant.
3: 3 Z Loc [ Ct        ]

(cont.)

Temperature:  Temperature is measured with a single instruction, P11, that
measures the thermistor resistance and calculates temperature.  Output is °C
when a multiplier of 1 and an offset of 0 is used.  See Section 10 of datalogger
manual for detailed information on the function of Instruction P11.

24:  Temp (107) (P11)
1: 1 Reps
2: 3 SE Channel
3: 2 Excite all reps w/E2
4: 4 Loc [ Temp_degC ]
5: 1.0 Mult
6: 0.0 Offset

(cont.)

5.3  Correction of Ionization Errors in EC Measurement
Ionization caused by the excitation of the EC sensor can cause large errors.
Campbell Scientific has developed a linear correction for conductivity between
0.005 and 0.44 mS cm-1, and a quadratic correction for conductivity between
0.44 and 7.0 mS cm-1.  Corrections were determined in standard salt solutions
containing KCl, Na2SO4, NaHCO3, and NaCl.

The following program segment automatically chooses which correction to
apply to the measurement.

;
;The following program set corrects for errors of ionization in the EC
;measurement.
;

25:  IF (X<=>F) (P89)
1: 3 X Loc [ Ct        ]
2: 4 <
3: .474 F
4: 30 Then Do

26:  Z=X*F (P37)
1: 3 X Loc [ Ct        ]
2: .95031 F
3: 3 Z Loc [ Ct        ]

27:  Z=X+F (P34)
1: 3 X Loc [ Ct        ]
2: -.00378 F
3: 3 Z Loc [ Ct        ]

28:  Else (P94)

29:  Polynomial (P55)
1: 1 Reps
2: 3 X Loc [ Ct        ]
3: 3 F(X) Loc [ Ct        ]
4: -.02889 C0
5: .98614 C1
6: .02846 C2
7: .000000 C3
8: .000000 C4
9: .000000 C5

30:  End (P95)
(cont.)

5.4  Correction of Temperature Errors
The effect of temperature on the sample solution can cause large errors in the
EC measurement.  A simple method of correcting for this effect is to assume a
linear relationship between temperature and EC.  This method generally
produces values to within 2% to 3% of a measurement made at 25°C.

The best corrections are made when the temperature coefficient is determined
at a temperature near field conditions.  See Section 9 for details on how to
determine the temperature coefficient.  If determining the temperature
coefficient is not possible, use a value of 2%/°C as a rough estimate.

The following program segment implements a previously determined
temperature coefficient (TC) and calculates the corrected conductivity.

;This next program set will correct errors in the EC measurement resulting
;from temperature differences.
;

31:  Z=X+F (P34)
1: 4 X Loc [ Temp_degC ]
2: -25 F
3: 6 Z Loc [ A         ]

32:  Z=X*F (P37)
1: 3 X Loc [ Ct        ]
2: 100 F
3: 7 Z Loc [ lOO_Ct    ]

33:  Z=X*F (P37)
1: 6 X Loc [ A         ]
2: nnn F Enter TC (%/°C) to correct cond. reading.
3: 8 Z Loc [ TC_Proces ]

34:  Z=X+F (P34)
1: 8 X Loc [ TC_Proces ]
2: 100 F
3: 8 Z Loc [ TC_Proces ]

35:  Z=X/Y (P38)
1: 7 X Loc [ lOO_Ct    ]
2: 8 Y Loc [ TC_Proces ]
3: 9 Z Loc [ C25mScm_l ] EC corrected for temperature.

(cont.)

5.5  Output Processing
Over large ranges, EC is not linear and is best reported as samples using
instruction P70.  In limited ranges, averaging (P71) measurements over time
may be acceptable.  Convention requires that the temperature at the time of the
measurement be reported.

;Output processing, convention states that the temperature be reported
;with the EC measurement.
;
36:  Do (P86)

1: 10 Set Output Flag High (Flag 0)

37:  Sample (P70)
1: 1 Reps
2: 3 Loc [ Ct        ]

38:  Sample (P70)
1: 1 Reps
2: 4 Loc [ Temp_degC ]

39:  Sample (P70)
1: 1 Reps
2: 9 Loc [ C25mScm_l ]

(cont.)

*Table 2 Program
02: 0.0 Execution Interval (seconds)

*Table 3 Subroutines

End Program (End)

6.  Calibration

6.1  Conversion Factors
1 S (Siemens) = 1 mho = 1/ohm

Although mS·cm-1 and µS·cm-1 are the commonly used units of EC, the SI base
unit is S·m-1.  The result of the example programs is mS·cm-1

EC measurements can be used to estimate dissolved solids.  For high accuracy,
calibration to the specific stream is required.  However, for rough estimates,
values between 550 and 750 mg·l-1 / mS·cm-1 are typical with the higher values
generally being associated with waters high in sulfate concentration (USGS
Water-Supply Paper #1473, p. 99).  A common practice is to multiply the EC in
mS·cm-1 by 500 to produce ppm or mg·l-1.

6.2  Typical Ranges
Single distilled water will have an EC of at least 0.001 mS·cm-1.  ECs of melted
snow usually range from 0.002 to 0.042 mS·cm-1.  ECs of stream water usually
range from 0.05 to 50.0 mS·cm-1, the higher value being close to the EC of sea
water (USGS Water-Supply Paper 1473, p. 102).

6.3  Factory Calibration
The CS547 is shipped with a cell constant calibrated in a 0.01 molal KCl
solution at 25.0°C ±0.05°C.  The solution has a EC of 1.408 mS cm-1.

6.4  Field Calibration
The cell constant is a dimensional number expressed in units of cm-1.  The unit
cm-1 is slightly easier to understand when expressed as cm·cm-2.  Because it is
dimensional, the cell constant as determined at any one standard, will change
only if the physical dimensions inside the CS547 probe change.  Error due to
thermal expansion and contraction is negligible.  Corrosion and abrasion,
however, have the potential of causing significant errors.

A field calibration of the CS547 cell constant can be accomplished as follows:

1. Make a 0.01 molal KCL solution by dissolving 0.7456 g of reagent grade
KCl in 1000 g of distilled water, or purchase a calibration solution.

2. Clean the probe thoroughly with the black nylon brush shipped with the
CS547 and a small amount of soapy water.  Rinse thoroughly with distilled
water, dry thoroughly, and place in the KCl solution.

3. Connect the CS547 and A547 or probe and interface to the datalogger
using the wiring described in Section 4.  Enter the following program into
the datalogger.

The calibration solution temperature must be between 1°C and 35°C; the
polynomial in step 11 (P58) corrects for temperature errors within this range.
The solution constant of 1.408 mS cm-1 (for prepared solution mentioned above),
entered in step 13 (P37), is valid only for a 0.01 molal KCl solution.  Location 8
[Kc (cm-1)], generated by step 14, will contain the resultant cell constant.

01:  AC Half Bridge (P5)
01: 1 Rep
02: 15 2500 mV fast Range (5000 mV fast for 21X)
03: 2 IN Chan
04: 1 Excite all reps w/EXchan 1
05: 2500 mV Excitation (5000 mV for 21X)
06: 1 Loc [Rs       ]
07: 1 Mult
08: 0 Offset

02:  BR Transform Rf[X/(1-X)] (P59)
01: 1 Rep
02: 1 Loc [Rs       ]
03: 1 Multiplier (Rf)

03:  Z=F (P30)
01: nnn F Enter Cable Length in Feet
02: 00 Exponent of 10
03: 5 Loc [Rp       ]

04:  Z=X*F (P37)
01: 5 Loc Rp
02: .00032 F
03: 5 Loc [Rp       ]

05:  Z=X*F (P37)
01: 5 Loc Rp
02: -.1 F
03: 5 Loc [Rp       ]

06:  Z=X+F (P34)
01: 5 Loc [Rp ]
02: -.005
03: 5 Loc [Rp       ]

07:  Z=X+Y (P33)
01: 1 X Loc Rs
02: 5 Y Loc Rp
03: 1 Z Loc [Rs       ]

08:  Temp 107 Probe (P11)
01: 1 Rep
02: 3 IN Chan
03: 2 Excite all reps w/EXchan 2
04: 2 Loc [t]
05: 1 Mult
06: 0 Offset

(cont.)

09:  Z=X+F (P34)
01: 2 X Loc t
02: -25 F
03: 3 Z Loc [(t-25).01]

10:  Z=X*F (P37)
01: 3 X Loc (t-25).01
02: .01 F
03: 3 Z Loc [(t-25).01]

11:  Polynomial (P55)
01: 1 Rep
02: 3 X Loc (t-25).01
03: 4 F(X) Loc [f(t)       ]
04: .99124 C0
05: -1.8817 C1
06: 3.4789 C2
07: -3.51 C3
08: -1.2 C4
09: -43 C5

12:  Z=1/X (P42)
01: 4 X Loc f(t)
02: 6 Z Loc [1/f(t)     ]

13:  Z=X*F (P37)
01: 6 X Loc 1/f(t)
02: 1.408 F EC of calibration solution
03: 7 Z Loc [Act'l Con]

14:  Z=X*Y (P36)
01: 7 X Loc Act'l Con
02: 1 Y Loc Rs
03: 8 Z Loc [Kc (cm-1)]

End

7.  Maintenance
Routine maintenance includes thoroughly cleaning the orifice of the CS547
probe with the black nylon brush provided and a little soapy water. Rinse
thoroughly.

8.  Analysis of Errors

8.1  EC Measurement Error
1. Bridge Measurement Error:  < 1.0%

2. Calibration Error:
bridge measurement: < 0.5%
calibration solution: < 1.0%

3. Ionization Error of KCl and Na+ Solutions After Correction:
< 2.0%, 0.45 to 7.0 mS cm-1

< 8.0%, 0.005 to 0.45 mS cm-1

Correction of Ionization Errors:  Figures 8.1-1 and 8.1-2 show the amount of
correction applied by the example program to compensate for ionization effects on
the measurements.  Also shown is an ideal correction.  Factors were derived by
measuring the standard solutions described in Section 2.2 with values of 0.0234,
0.07, 0.4471, 07, 1.413, 2.070, 3.920, and 7.0 mS cm-1.

FIGURE 8.1-1.  Plot of Ideal and Actual Correction between
0 and 0.44 mS cm-1

FIGURE 8.1-2.  Plot of Ideal and Actual Correction between
0.44 and 7.0 mS cm-1

8.2  Temperature Measurement Error
The overall probe accuracy is a combination of the thermistor's interchangeability
specification, the precision of the bridge resistors, and the polynomial error.  In a
"worst case" all errors add to an accuracy of ±0.4°C over the range of -24° to 48°C
and ±0.9°C over the range of -38°C to 53°C.  The major error component is the
interchangeability specification of the thermistor, tabulated in Table 8.2-1.  For the
range of 0° to 50°C the interchangeability error is predominantly offset and can be
determined with a single point calibration.  Compensation can then be done with an
offset entered in the measurement instruction.  The bridge resistors are 0.1%
tolerance with a 10 ppm temperature coefficient.  Polynomial errors are tabulated in
Table 8.2-2 and plotted in Figure 8.2-1.

TABLE 8.2-1.  Thermistor
Interchangeability Specification

Temperature

Temperature (°C) Tolerance (±°C)
−40 0.40

−30 0.40

−20 0.32

−10 0.25

0 to +50 0.20

TABLE 8.2-2.  Polynomial Error

-40 to +56 <±1.0°C
-38 to +53 <±0.5°C
-24 to +48 <±0.1°C

FIGURE 8.2-1.  Error Produced by Polynomial Fit to Published Values

9.  Deriving a Temperature Compensation Coefficient
1. Place the CS547 in a sample of the solution to be measured.  Bring the

sample and the probe to 25°C.

2. Enter the example program from Section 5.2 in the datalogger and record
Ct at 25°C from Location 3.  This number will be C25 in the formula in
Step 4.

3. Bring the solution and the probe to a temperature (t) near the temperature
at which field measurements will be made.  This temperature will be t (in
°C) in the formula.  Record Ct at the new temperature from Location 3.
This number will be C in the formula in Step 4.

4. Calculate the temperature coefficient (TC) using the following formula.

TC
C C

t C
C= ∗

−
− ∗

°100
25

25

25

( )
( )

%/=

Enter TC in the appropriate location (nnn) as shown in the program segment in
Section 5.4.

10.  Instruction 11 Details
Understanding the details in this section are not necessary for general operation
of the CS547 probe with CSI's dataloggers.

Instruction 11 outputs a precise 2 VAC excitation (4 V with the 21X) and
measures the voltage drop due to the sensor resistance.  The thermistor
resistance changes with temperature.  Instruction 11 calculates the ratio of
voltage measured to excitation voltage (Vs/Vx) which is related to resistance,
as shown below:

Vs/Vx = 1000/(Rs+249000+1000)

where Rs is the resistance of the thermistor.

See the measurement section of the datalogger manual for more information on
bridge measurements.

Instruction 11 then calculates temperature using a fifth order polynomial equation
correlating Vs/Vx with temperature.  The polynomial coefficients are given in
Table 10-2. The polynomial input is (Vs/Vx)*800.  Resistance and datalogger
output at several temperatures are shown in Table 10-1.

TABLE 10-1.  Temperature , Resistance, and
Datalogger Output

0.00 351017 -0.06
2.00 315288 1.96
4.00 283558 3.99
6.00 255337 6.02
8.00 230210 8.04

10.00 207807 10.06
12.00 187803 12.07
14.00 169924 14.06
16.00 153923 16.05
18.00 139588 18.02
20.00 126729 19.99
22.00 115179 21.97
24.00 104796 23.95
26.00 95449 25.94
28.00 87026 27.93
30.00 79428 29.95
32.00 72567 31.97
34.00 66365 33.99
36.00 60752 36.02
38.00 55668 38.05
40.00 51058 40.07
42.00 46873 42.07
44.00 43071 44.05
46.00 39613 46.00
48.00 36465 47.91
50.00 33598 49.77
52.00 30983 51.59
54.00 28595 53.35
56.00 26413 55.05
58.00 24419 56.70
60.00 22593 58.28

TABLE 10-2.  Polynomial
Coefficients

COEFFICIENT VALUE

C0 -53.4601
C1 9.08067
C2 -8.32569 x 10-01

C3 5.22829 x 10-02

C4 -1.67234 x 10-03

C5 2.21098 x 10-05

11.  Electrically Noisy Environments
AC power lines can be the source of electrical noise.  If the datalogger is in an
electronically noisy environment, the 107/107B temperature measurement
should be measured with the AC half bridge (Instruction 5) with the 60 Hz
rejection integration option on the CR10(X) and slow integration on the 21X
and CR7 (see Section 13 of the datalogger manual for more information on
noise).  Instruction 11's fast integration will not reject 60 Hz noise.

Example 2.  Sample CR10(X) Instructions Using AC Half Bridge

01:  AC Half Bridge (P5)
01: 1 Rep
02: 22** 7.5 mV 60 Hz rejection Range
03: 9* IN Chan
04: 3* Excite all reps w/EXchan 3
05: 2000** mV Excitation
06: 11* Loc [:Air_Temp ]
07: 800 Mult
08: 0 Offset

02:  Polynomial (P55)
01: 1 Rep
02: 11* X Loc Air_Temp
03: 11* F(X) Loc [:Air_Temp ]
04: -53.46 C0
05: 90.807 C1
06: -83.257 C2
07: 52.283 C3
08: -16.723 C4
09: 2.211 C5

* Proper entries will vary with program and datalogger channel and input location
assignments.

**On the 21X and CR7 use the 15 mV input range and 4000 mV excitation.

12.  Long Lead Lengths Temperature
If the CS547 has lead lengths of more than 300 feet, use the DC Half Bridge
instruction (Instruction 4) with a 2 millisecond delay to measure temperature.
The delay provides a longer settling time before the measurement is made.  Do
not use the CS547 with long lead lengths in an electrically noisy environment.

Example 3.  Sample Program CR10 Using DC Half Bridge with Delay

01:  Excite, Delay,Volt(SE) (P4)
01: 1 Rep
02: 2** 7.5 mV slow range
03: 9* IN Chan
04: 3* Excite all reps w/EXchan 3
05: 2 Delay (units .01sec)
06: 2000** mV Excitation
07: 11* Loc [:Temp_C   ]
08: .4*** Mult
09: 0 Offset

(cont.)

02:  Polynomial (P55)
01: 1 Rep
02: 11* X Loc Temp_C
03: 11* F(X) Loc [:Temp_C   ]
04: -53.46 C0
05: 90.807 C1
06: -83.257 C2
07: 52.283 C3
08: -16.723 C4
09: 2.211 C5

* Proper entries will vary with program and datalogger channel and input location
assignments.

** On the 21X and CR7 use the 15 mV input range and 4000 mV excitation.
*** Use a multiplier of 0.2 with a 21X and CR7.

13.  CS547 Schematic

Black (Ex Cond)

Green (Ex Temp)

Red (Temp)

Orange (Cond)

FIGURE 13-1.  CS547 Conductivity and Temperature Circuit Diagram

FIGURE 13-2.  A547 Interface Circuit Diagram

